
Regressione

(* Definiamo innanzi tutto le funzioni per calcolare lo scarto
   quadratico totale e la deviazione standard dei dati rispetto
   ai punti calcolati mediante l'espressione esp nella variabile x *)

In [1 ] :=
ScartoQuadTotale[dati_,esp_,x_] :=
  Block[{td = Transpose[dati]},
    Apply[Plus,((esp /. x -> td[[1]]) - td[[2]])^2]]
    
  (* trasponendo la matrice dei dati separiamo le x dalle y ... *)
  (* la sostituzione esp /. x -> td[[1]] rappresenta il modo più
     semplice per ottenere la lista dei valori calcolati mediante
     l'espressione esp, grazie al fatto che tutte le funzioni
     matematiche elementari sono implementate nel sistema in modo
     da funzionare anche con liste di valori *)

In [5 ] :=
DeviazioneStandard[dati_,esp_,x_] := 
  N[Sqrt[ScartoQuadTotale[dati,esp,x]/Length[dati]]]

(* Regressione lineare *)

In [6 ] :=
Regressione[dati_,x_] :=
  Block[{m$,q$},
    Block[{SQT$ = ScartoQuadTotale[dati,m$ x + q$,x]},
      m$ x + q$ /. Solve[{D[SQT$,m$] == 0,D[SQT$,q$] == 0},{m$,q$}][[1]]]]
      
  (* il primo Block ha il solo scopo di localizzare le variabili m$ e q$
     il secondo Block serve anche ad assegnare il valore a SQT$ (in tal
     modo lo scarto quadratico viene calcolato una sola volta anche se
     poi è utilizzato due volte) *)
  (* si assume che il carattere $ non sia normalmente usato nei nomi
     dei simboli quindi m$, q$ e SQT$ non dovrebbero sovrapporsi ad
     altri simboli *)

(* controlliamo che funzioni ... *)

In [10 ] :=
dati = Table[{x,.5 x + Random[Real,{-.1,.1}]},{x,0.,1.,.005}];

In [11 ] :=
punti = ListPlot[dati,AspectRatio->Automatic]

Out[11]=
-Graphics-

In [12 ] :=
retta = Regressione[dati,x]

Out[12]=
-0.00796042 + 0.508912 x

(* verifichiamo con la funzione Fit ... *)

In [13 ] :=
Fit[dati,{1,x},x]

Out[13]=
-0.00796042 + 0.508912 x

In [14 ] :=
dev = DeviazioneStandard[dati,retta,x]

Out[14]=
0.0565601

In [15 ] :=
Plot[retta,{x,0,1},AspectRatio->Automatic]

-Graphics-

In [16 ] :=
Show[%,punti,AspectRatio->Automatic]

-Graphics-

(* Regressione generalizzata = funzione Fit *)

In [17 ] :=
RegresGen[dati_,esp_,x_] :=
  Block[{k$,l$ = Length[esp],e$ = N[esp]},
    Block[{SQT$ = ScartoQuadTotale[dati,Sum[k$[i] e$[[i]],{i,1,l$}],x]},
      Sum[k$[i] esp[[i]],{i,1,l$}] /. 
          NSolve[Table[D[SQT$,k$[i]] == 0,{i,1,l$}],
                 Table[k$[i],{i,1,l$}]][[1]] ]]
                    
  (* esp è la lista delle espressioni da combinare linearmente,
     k$[1],k$[2],...,k$[l$] sono i coefficienti per la comb. lineare *)

In [20 ] :=
dati = Table[{x,.5 x^2 + .2 Sin[2 N[Pi] x] + .1 Sin[6 N[Pi] x] +
                Random[Real,{-.05,.05}]},
             {x,0.,1.,.005}];

In [21 ] :=
punti = ListPlot[%,AspectRatio->Automatic]

0.2 0.4 0.6 0.8 1

0.1

0.2

0.3

0.4

0.5

Out[21]=
-Graphics-

In [22 ] :=
retta = RegresGen[dati,{1,x},x]

Out[22]=
0.143392 + 0.0513107 x

In [23 ] :=
DeviazioneStandard[dati,retta,x]

Out[23]=
0.100412

In [24 ] :=
Plot[{retta,retta - dev,retta + dev},{x,0,1},AspectRatio->Automatic]

In [25 ] :=
Show[%,punti,AspectRatio->Automatic]

In [26 ] :=
parabola = RegresGen[dati,{1,x,x^2},x]

Out[26]=
                                 2
0.229308 - 0.46678 x + 0.518091 x

In [27 ] :=
dev = DeviazioneStandard[dati,parabola,x]

Out[27]=
0.0925281

In [28 ] :=
Plot[{parabola,parabola - dev,parabola + dev},{x,0,1},
     AspectRatio->Automatic]

-Graphics-

In [29 ] :=
Show[%,punti,AspectRatio->Automatic]

In [30 ] :=
curva = RegresGen[dati,{1,x,x^2,Sin[2 Pi x],Sin[4 Pi x],Sin[6 Pi x]},x]

Out[30]=
                                    2
0.0173966 - 0.0429567 x + 0.518091 x  + 0.193784 Sin[2 Pi x] + 
 
  0.00150024 Sin[4 Pi x] + 0.092275 Sin[6 Pi x]

In [31 ] :=
Fit[dati,{1,x,x^2,Sin[2 Pi x],Sin[4 Pi x],Sin[6 Pi x]},x]

Out[31]=
                                    2
0.0173966 - 0.0429567 x + 0.518091 x  + 0.193784 Sin[2 Pi x] + 
 
  0.00150024 Sin[4 Pi x] + 0.092275 Sin[6 Pi x]

In [32 ] :=
dev = DeviazioneStandard[dati,curva,x]

Out[32]=
0.0275691

In [33 ] :=
Plot[{curva,curva - dev,curva + dev},{x,0,1},
     AspectRatio->Automatic]

-Graphics-

In [34 ] :=
Show[%,punti,AspectRatio->Automatic]

